MLC Lab Visit - Lab 02 - Maple

Mth 355 (a.k.a. Mth 399) Jan 15, 2003 Maple 7
Revised slightly Feb 11, 2003
Bent E. Petersen

There are 3 problems below. Problem solutions are due Jan 22, 2003. Email your solutions to me as Maple worksheet attachments. Your worksheet must execute correctly for full credit.

In this worksheet we investigate a few more features of Maple.

If you are viewing the MWS version of this document you will note all the Maple output has been removed. You will have to execute each command (by pressing Enter when the cursor is on the command line) to see the output. Take the opportunity to experiment! Change some things.

The static PDF version of this document shows all of the Maple output. It is useful to look at when Maple is not available.

```
> restart;
```

Digits

You set the Maple's floating point precision by assigning a value to Digits (the default is 10). Maple usually does exact calculations, but when floating point numbers are involved then Digits sets the precision. Here's an amusing example

```
> Digits:=4: convert(evalf(Pi),rational);
```

$$\frac{22}{7}$$

The conversion to a rational number makes use of Digits, rather than any precision specified in the evalf() command. You can easily find other rational approximations to pi

```
> Digits:=8: convert(evalf(Pi),rational);
```

$$\frac{355}{113}$$

The label "rational" is protected in Maple 6. You can not assign a value to it (which is just as well).

Let's set Digits back to its default.

```
> Digits:=10:
>
```

Functions and Expressions

Maple distinguishes between functions and expressions. Here's one way to define a function:

```
> f:=x->sin(3*x+x^2);
```

$$f := x \rightarrow \sin(3\,x + x^2)$$

We can also define an expression:

```
> g:=sin(3*x+x^2);
```

$$g := \sin(3\,x + x^2)$$

Both of the examples above assume that x has not already been assigned a value. It needs to be an unassigned variable. In the definition of f the x is a dummy variable, a place marker. In g however, it is part of the expression, and one can refer to it.

To evaluate a function we use the usual function convention. To evaluate an expression one generally uses the subs() command (though it has other subtle uses).

```
> f(1); subs(x=1,g);
```

$$\sin(4)$$

$$\sin(4)$$

Note the subs() command above does not assign a value to x.

An expression can also be evaluated by using the eval() command, but do check help to make sure you don't have any surprises in more complicated situations. The commands eval() and subs() work in quite different ways. In the simple case that we illustrated here eval() is actually the preferred command to use.

```
> eval(g,x=1);
```

$$\sin(4)$$

Note the eval() command above does not assign a value to x.

We can convert an expression into a function by using the unapply() command

```
> h:=unapply(g,x);
```

$$h := x \rightarrow \sin(3x + x^2)$$

You can think of unapply() as turning the indicated variable(s) into dummy variables or place markers. Thus f(x) is the the function f evaluated at x and unapply(f(x),x) ought to return the function f. Let's check that:

```
> ff:=unapply(f(x),x); (ff-f)(w);
```

$$ff := x \rightarrow \sin(3x + x^2)$$

$$0$$

Sure enough!

If you have an inquisitive nature you probably wonder if Maple has an apply() command. It does but the functional notation is usually more convenient.

```
> is(f(t) = apply(f,t));
```

$$true$$

```
>
```

Derivatives

Some Maple commands work on expressions, some work on functions, and some on both. For example, here are the derivatives of f and g (defined above).

```
> D(f); diff(g,x);
```

$$x \rightarrow \cos(3x + x^2)(3 + 2x)$$

$$\cos(3x + x^2)(3 + 2x)$$

Second derivatives are no problem

```
> D(D(f)); diff(g,x,x);
```

$$x \rightarrow -\sin(3x + x^2)(3 + 2x)^2 + 2\cos(3x + x^2)$$

$$-\sin(3x + x^2)(3 + 2x)^2 + 2\cos(3x + x^2)$$

but this notation can get out hand. Fortunately there is an alternative! Here are the fourth derivatives as an illustration:

```
> (D@@4)(f); diff(g,x$4);
```

$$x \rightarrow \sin(3x + x^2)(3 + 2x)^4 - 12\cos(3x + x^2)(3 + 2x)^2 - 12\sin(3x + x^2)$$

$$\sin(3x + x^2)(3 + 2x)^4 - 12\cos(3x + x^2)(3 + 2x)^2 - 12\sin(3x + x^2)$$

Partial derivatives of expressions are also easily computed (here once relative to y and three times relative to x):

```
> diff(x/(x^2+y^2),x$3,y);
```

$$-288\frac{x^2 y}{(x^2+y^2)^4}+\frac{24 y}{(x^2+y^2)^3}+\frac{384 x^4 y}{(x^2+y^2)^5}$$

There is an inert version Diff() of diff(). An inert function returns unevaluated. That may seem strange, but sometimes one can save time by postponing evaluation, or one can prevent Maple from attempting a calculation that will fail at present, but can be carried out later in special cases or different contexts. Unevaluated expressions may be evaluated by using the command value(), thought there are other ways.

Inert functions, together with the ditto operator can be used to get nicely typeset expressions. See if you can sort out the following:

```
> Diff(x/(x^2+y^2),x$3,y): %=value(%);
```

$$\frac{\partial^4}{\partial y\,\partial x^3}\frac{x}{x^2+y^2}=-288\frac{x^2 y}{(x^2+y^2)^4}+\frac{24 y}{(x^2+y^2)^3}+\frac{384 x^4 y}{(x^2+y^2)^5}$$

In Maple we can label just about any expression. Thus we can achieve the effect above in a manner that may seem more natural:

```
> expr01:=x/(x^2+y^2),x$3,y: Diff(expr01)=diff(expr01);
```

$$\frac{\partial^4}{\partial y\,\partial x^3}\frac{x}{x^2+y^2}=-288\frac{x^2 y}{(x^2+y^2)^4}+\frac{24 y}{(x^2+y^2)^3}+\frac{384 x^4 y}{(x^2+y^2)^5}$$

Note equals sign is just part of the expression. It is not an assignment.

```
>
```

Integration

Let's bring back some fond memories from calculus - the problem of integration. Here's an example to get you started:

```
> int(1/(1+x^4),x);
```

$$\frac{1}{8}\sqrt{2}\ \ln\left(\frac{x^2+x\sqrt{2}+1}{x^2-x\sqrt{2}+1}\right)+\frac{1}{4}\sqrt{2}\ \arctan(x\sqrt{2}+1)+\frac{1}{4}\sqrt{2}\ \arctan(x\sqrt{2}-1)$$

Again you can use postponed evaluation to obtain a nicely typeset expression

```
> Int(1/(1+x^4),x): %=value(%);
```

$$\int\frac{1}{1+x^4}\,dx=\frac{1}{8}\sqrt{2}\ \ln\left(\frac{x^2+x\sqrt{2}+1}{x^2-x\sqrt{2}+1}\right)+\frac{1}{4}\sqrt{2}\ \arctan(x\sqrt{2}+1)+\frac{1}{4}\sqrt{2}\ \arctan(x\sqrt{2}-1)$$

$$\int\frac{1}{1+x^4}\,dx=\frac{1}{8}\sqrt{2}\ \ln\left(\frac{x^2+x\sqrt{2}+1}{x^2-x\sqrt{2}+1}\right)+\frac{1}{4}\sqrt{2}\ \arctan(x\sqrt{2}+1)+\frac{1}{4}\sqrt{2}\ \arctan(x\sqrt{2}-1)$$

Note that int() function works on expressions, not functions. Thus to integrate a function you have to convert it to an expression by evaluating it at a dummy variable.

```
> h:=t->t^3; int(h(u),u);
```

$$h := t \to t^3$$

$$\frac{1}{4}u^4$$

I bet you wish you had a tool like this when you were studying calculus!

Naturally definite integrals are possible too, even some improper integrals.

```
> exp(-t^2),t=0..infinity: Int(%)=int(%);
```

$$\int_0^{\infty}\mathbf{e}^{(-t^2)}\,dt=\frac{1}{2}\sqrt{\pi}$$

If you want a floating point number you can simply use evalf(), but there is a subtle and important difference depending on how you do it.

```
> a:=int(2*x^2*log(x)^3+x^3*log(x),x=1..2): evalf(a,16);
```

$$2.476290396904212$$

```
> evalf(Int(2*x^2*log(x)^3+x^3*log(x),x=1..2),16);
```

$$2.476290396904210$$

In the first case we assign the symbolic expression for the integral to a and then evaluate that expression. In the second example, Maple detects that we want a numeric result and evaluates the integral numerically without first trying to obtain a symbolic solution. This is important. For example

```
> restart;
> stime:=time():
> int(arctan(x)/log(x),x=Pi/8..Pi/4); evalf(%);
```

$$\int_{1/8\,\pi}^{1/4\,\pi} \frac{\arctan(x)}{\ln(x)}\,dx$$

$$-.4623890373$$

```
> etime:=time()-stime: 'time'=etime;
```

$$time = 7.813$$

```
>
> restart;
> stime:=time():
> evalf(Int(arctan(x)/log(x),x=Pi/8..Pi/4));
```

$$-.4623890373$$

```
> etime:=time()-stime: 'time'=etime;
```

$$time = .234$$

Here, in the first case, Maple decided after a while (possibly a long while) that it can not return a symbolic value for the integral and so returned it unevaluated. Then evalf() called a numeric quadrature rule to get an answer. In the second case however, Maple wasted no time trying to find a nonexistent symbolic solution, but instead used a numeric quadrature method. This is an important use of inert functions. You can grow noticably older waiting for a symbolic solution to a complex problem.

There are refinements. For example, you can specify what quadrature method to use. Enter the command ?int[numeric] for more informatio

```
>
```

Plotting

Functions and expressions can be plotted. There are numerous plot variations. Check the help facility, ?plot, for details.

```
> f:=x->sin(1/x); g:=sin(1/x);
```

$$f := x \to \sin\left(\frac{1}{x}\right)$$

$$g := \sin\left(\frac{1}{x}\right)$$

```
> plot(f,0..1,numpoints=200,title="Plotting a function");
```




```
> plot(g,x=0..1,numpoints=200,title="Plotting an expression"):
```

Replace the colon above by semicolons and press Enter to generate the plot. Of course, it will look the same as the previous plot.

We can convert a function into an expression simply by evaluating it, so one can also do

```
>  plot(f(x),x=0..1):
```

Replace the colon above by a semicolon and press Enter to generate the plot.

You can also plot anonymous functions, or expressions, that is, plot them without first assigning them to a variable:

```
> plot(x->x+sin(x),0..4*Pi);
```




```
> plot(x+sin(x),x=0..4*Pi):
```

Replace the colon above by semicolons and press Enter to generate the plot.

Maple has many plot types. Explore Maple help to see if you can find what you want.

Here is a well-know plot.

```
> plot3d(sin(sqrt(x^2+y^2))/sqrt(x^2+y^2),x=-7..7,y=-7..7);
```



Note you can use the mouse to grab the plot and rotate it to view the surface from different angles. You can also specify the initial default orientation as part of the plot3d command. Check ?plot3d[option]

We can also do parametric plots. We will use parameters t and p, so lets make sure first they have not been assigned to some other expressions (otherwise we will get incomprehensibe error messages).

```
> t:=evaln(t): p:=evaln(p):
> plot3d([4*cos(t)*sin(p),4*sin(t)*sin(p),4*cos(p)],t=-Pi..Pi/2,p
  =0..Pi/2);
```



Note again you can drag the plot around with the mouse to see the surface from different view points.

Contour plots are often useful for studying functions of 2 variables:

```
> contourplot(x^2-18*x*y+4*y^2,x=-10..10,y=-15..15,thickness=2,color=blue);
```

$$contourplot(x^2 - 18\,x\,y + 4\,y^2, x = -10 .. 10, y = -15 .. 15, thickness = 2, color = blue)$$

You can even specify the contours (very handy):

```
> contourplot(x^2-18*x*y+4*y^2,x=-10..10,y=-15..15,thickness=2,color=blue,contours=[-230,-160,0,160,230]);
```

$$contourplot(x^2 - 18\,x\,y + 4\,y^2, x = -10 .. 10, y = -15 .. 15, thickness = 2, color = blue, contours = [-230, -160, 0, 160, 230])$$

Many other plot commands are made available by loading the plots package - with(plots).

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> tubeplot([t,t^2,t*sin(t)],t=-1..22,radius=6*(2+cos(t/4)));
```



```
> tubeplot([4*cos(t),4*sin(t),4*t],t=0..12,radius=3);
```



```
> tubeplot([t,t^2,t^3],t=0..1,radius=1);
```



```
> spacecurve([t,t^2,t^3],t=0..1,thickness=3,color=blue,axes=boxed
  );
```



Try rotating this plot with the mouse to see if you can make sense out of it.

```
>
```

Stirling Numbers of the second kind, S(r,n)

```
> restart;
> with(combinat):
Warning, the protected name Chi has been redefined and unprotected
```

S(r,n), denoted by stirling2(r,n) in Maple, is the number of partitions of a set of cardinality r into n (nonempty) subsets. Clearly we must have $n \le r$. You can think of it as the number of ways to place r distinguishable objects into n indistinguishable buckets with no bucket empty. By convention there is one partition of the empty set (the empty partition) with no subsets.

```
> S(0,0) = stirling2(0,0); S(4,0) = stirling2(4,0); S(4,2) =
  stirling2(4,2); S(7,3) = stirling2(7,3); S(21,4) =
  stirling2(21,6);
```

$$S(0, 0) = 1$$

$$S(4, 0) = 0$$

$$S(4, 2) = 7$$

$$S(7, 3) = 301$$

$$S(21, 4) = 26585679462804$$

The last two entries above show that a direct by-hand count may be difficult or infeasible.

```
>
```

Bell Numbers

```
> restart;
> with(combinat):
Warning, the protected name Chi has been redefined and unprotected
```

The Maple function call bell(n) returns the n-th Bell number. The n-th Bell number is the number of ways that a set with n elements can be partitioned into a union of disjoint (nonempty) subsets. For example, the set {1,2,3,4} and be partioned in the following ways:

{{1},{2},{3},{4}}
{{1,2},{3},{4}}
{{2,3},{1},{4}}
{{3,4},{1}{2}}
{{1,4}.{2},{3}}
{{1,3},{2},{4}}
{{1,2},{3,4}}
{{2,3},{4,1}}

{{3,4},{1,2}}
{{1,4},{2,3}}
{{1,2,3},{4}}
{{2,3,4},{1}}
{{3,4,1},{2}}
{{4,2,1},{3}}
{{1,2,3,4}}

so in 15 ways. Let's see what Maple says

```
> bell(4);
```

$$15$$

The Bell numbers grow fairly quickly with n,

```
> 'bell(8)' = bell(8); 'bell(10)' = bell(10); 'bell(20)' =
  bell(20);
```

$$\text{bell}(8) = 4140$$

$$\text{bell}(10) = 115975$$

$$\text{bell}(20) = 51724158235372$$

Clearly a direct enumeration like the one we did above rapidly becomes impractical. From the definition it is clear that

```
> 'bell(r)' = Sum('S(r,n)',n=0..r);
```

$$\text{bell}(r) = \sum_{n=0}^{r} \text{S}(r, n)$$

Let's check this relation for the case enumerated above:

```
> sum(stirling2(4,n),n=0..4);
```

$$15$$

There are a number of interpretations of Bell numbers. For example, consider the problem of putting r distinguishable objects into r indistinguishable buckets (in no particular order, since otherwise we could distinguish them), with some buckets possibly left empty. Each arrangement (if we ignore the empty buckets) may be considered a partition of the set of objects. Thus the number of ways is bell(n).

```
>
```

Number of factorizations of a square-free natural number

A natural number is square-free if it has no square factors. Such a number is a product of distinct primes. In this case we can find all factorization by partitioning the set of prime factors and multiplying out the ones corresponding to subsets in the partition. Thus the number of factorizations is bell(r) where r is the number of (distinct) prime factors.

For example, 210 is the product of the primes 2, 3, 5 and 7 and so has bell(4) = 15 factorizations.

\>

Construction of all partitions

Sometimes we want an explicit list of all partitions of a set so we can perform some operation for each partition by stepping through the list. Here is a procedure which returns the list of all partitions of the set {1,2,3,....,n}.

```
> parts:=proc(n)
>   local  P,A,B,C; option remember;
> if n=1 then
>   P:=[{1}];
> elif n=2 then
>   P:=[ {{1},{2}}, {{1,2}} ];
> elif n>2 then
>   P:=[];
>   for A in parts(n-1) do
>     P:= [ op(P),  A union {{n}} ];
>     for B in A do
>       C:= ( A minus {B} ) union { B union {n} };
>       P:= [op(P), C ];
>     od;
>   od;
> elif n=0 then
>   P:= [];
> else
>   P:=FAIL;
> fi;
> return P;
> end:
```

Let’s check that it works:

```
> parts(3);
```

[{{1}, {2}, {3}}, {{2}, {1, 3}}, {{1}, {2, 3}}, {{1, 2}, {3}}, {{1, 2, 3}}]

Be careful, parts(n) grows rapidly with n. If you want to experiment save your work first.

```
>
```

r-permutations

The procedure permute(m,r) returns a list of all r-permutations of {1,2,..,m}. The procedure numbperm(m,r) returns the number of r-permutations. Thus

```
> permute(5,3); numbperm(5,3);
```

[[1, 2, 3], [1, 2, 4], [1, 2, 5], [1, 3, 2], [1, 3, 4], [1, 3, 5], [1, 4, 2], [1, 4, 3], [1, 4, 5], [1, 5, 2], [1, 5, 3], [1, 5, 4], [2, 1, 3], [2, 1, 4], [2, 1, 5], [2, 3, 1], [2, 3, 4], [2, 3, 5], [2, 4, 1], [2, 4, 3], [2, 4, 5], [2, 5, 1], [2, 5, 3], [2, 5, 4], [3, 1, 2], [3, 1, 4], [3, 1, 5], [3, 2, 1], [3, 2, 4], [3, 2, 5], [3, 4, 1], [3, 4, 2], [3, 4, 5], [3, 5, 1], [3, 5, 2], [3, 5, 4], [4, 1, 2], [4, 1, 3], [4, 1, 5], [4, 2, 1], [4, 2, 3], [4, 2, 5], [4, 3, 1], [4, 3, 2], [4, 3, 5], [4, 5, 1], [4, 5, 2], [4, 5, 3], [5, 1, 2], [5, 1, 3], [5, 1, 4], [5, 2, 1], [5, 2, 3], [5, 2, 4], [5, 3, 1], [5, 3, 2], [5, 3, 4], [5, 4, 1], [5, 4, 2], [5, 4, 3]]

60

Like most Maple functions permute() is polymorphic. If we pass a list L, say permute(L,r), we get all the r-permutaions of the elements of the list L

```
> permute([a,b,c,d],3); numbperm([a,b,c,d],3);
```

[[a, b, c], [a, b, d], [a, c, b], [a, c, d], [a, d, b], [a, d, c], [b, a, c], [b, a, d], [b, c, a], [b, c, d], [b, d, a], [b, d, c], [c, a, b], [c, a, d], [c, b, a], [c, b, d], [c, d, a], [c, d, b], [d, a, b], [d, a, c], [d, b, a], [d, b, c], [d, c, a], [d, c, b]]

24

Note permute() and numbperm() even work correctly on lists with repeated elements

```
> permute([a,b,c,c],3); numbperm([a,b,c,c],3);
```

[[a, b, c], [a, c, b], [a, c, c], [b, a, c], [b, c, a], [b, c, c], [c, a, b], [c, a, c], [c, b, a], [c, b, c], [c, c, a], [c, c, b]]

12

```
>
```

Multinomial and Binomial Coefficients

The Maple functions multinomial() and binomial =() return what you expect:

```
> binomial(9,4);
```

$$126$$

```
> multinomial(8,3,3,2);
```

$$560$$

```
> expand((x+y+z)^6);
```

$$60\,x\,y^3\,z^2 + z^6 + 15\,x^2\,y^4 + 6\,x^5\,z + 15\,x^4\,y^2 + 15\,x^4\,z^2 + 20\,x^3\,y^3 + 15\,x^2\,z^4 + 6\,x^5\,y + 6\,x\,y^5 + 6\,x\,z^5 + 6\,y^5\,z + 15\,y^4\,z^2 + 20\,y^3\,z^3 + 15\,y^2\,z^4 + 6\,y\,z^5 + 30\,x^4\,y\,z + 20\,x^3\,z^3 + 90\,x^2\,y^2\,z^2 + 60\,x^3\,y^2\,z + 60\,x^3\,y\,z^2 + 60\,x^2\,y^3\,z + 60\,x^2\,y\,z^3 + 30\,x\,y\,z^4 + x^6 + y^6 + 30\,x\,y^4\,z + 60\,x\,y^2\,z^3$$

The calculation

```
> multinomial(6,2,1,3); multinomial(6,2,2,2);
```

$$60$$

$$90$$

illustrates two terms of the multinomial expansion (x1+x2+...+xn)^m = sum(multinomial(m,k1...kn)* x1^k1* x2^k2* ....*xn^kn, k1+k2+...+kn = m).

```
>
```

r-combinations

We can select r objects, without regard to order, from a set of n objects (distinguishable since elements of a set) in C(n,r) = P(n,r)/r! = n!/(r!(n-r)!) ways, because P(n,r) is the number of ways to choose the r objects in order, and if we disregard the order, we clearly have to divide by r!. C(n,r) is called binomial(n,r) in Maple. It is the number of combinations of n distinct objects taken r at a time.

```
> binomial(12,4);
```

$$495$$

Note Maple can even simplify some expressions involving binomial coefficients

```
> sum(binomial(n,k)^2,k=0..n); convert(%,factorial);
```

$$\frac{\Gamma(2\,n+1)}{\Gamma(n+1)^2}$$

$$\frac{(2\,n+1)!\,(n+1)^2}{(2\,n+1)\,((n+1)!)^2}$$

Here Maple resists simplifying the factorials, but the answer is clear.

>

Problems

Problem 1

A partition of a set X is a (special) subset of P(X). Hence if |X| = n the number of partitions of X is bounded by 2^(2^n), and is probably much smaller. Conjecture log(bell(floor(x))) <= C 2^x, x>=1, and use Maple to estimate the constant C.

(Yes. 10^9 is a valid answer, but I want a smaller estimate.)

Problem 2

Find the number of factorizations of 23205. Note

```
> ifactor(23205);
```

$$(3)\,(5)\,(7)\,(13)\,(17)$$

>

Can you coerce Maple into listing all of the factorizations, for example, by stepping over the list of partitions of a suitable set?

Problem 3

A small college has 3 dormitories, imaginatively called dorm A, dorm B and dorm C. Each dormitory has a large number of vacancies when 60 new students arrive on campus.

Part (A). How many ways can we distribute the 60 students among the 3 distinguishable dorms if we assume the students are indistinguishable, that is, we just care how many students end up in each dorm?

Part (B). How many ways can we distribute the 60 students among the 3 distinguishable dorms if we assume the students are indistinguishable, that is, we just care how many students end up in each dorm, and if we require that each dorm gets at least one student?

>
>